Maple-assisted proof of formula for A297301

Robert Israel

25 May 2018

There are $2^8 = 256$ configurations for a 4×2 sub-array. Consider the 256×256 transition matrix T such that $T_{ij} = 1$ if the left two columns of a 4×3 sub-array could be in configuration i while the right two columns are in configuration j (i.e. the middle column is compatible with both i and j, and every 1 in the middle column has two horizontal or antidiagonal neighbours that are 1), and 0 otherwise. The following Maple code computes it. Configurations are encoded as 8-element lists in the order

$$\begin{bmatrix} 1 & 5 \\ 2 & 6 \\ 3 & 7 \\ 4 & 8 \end{bmatrix}$$

```
> Configs:=  [seq(convert(2^8+i,base,2)[1..8],i=0..2^8-1)]:
> Compatible:= proc(i,j)
    if Configs[i][5..8] <> Configs[j][1..4] then return 0 fi;
    if Configs[i][5] = 1 and Configs[i][1]+Configs[i][2]+Configs[j]
  [5] <> 2 then return 0 fi;
    if Configs[i][6] = 1 and Configs[i][2]+Configs[i][3]+Configs[j]
  [5]+Configs[j][6] <> 2 then return 0 fi;
    if Configs[i][7] = 1 and Configs[i][3]+Configs[i][4]+Configs[j]
  [6]+Configs[j][7] <> 2 then return 0 fi;
    if Configs[i][8] = 1 and Configs[i][4]+Configs[j][7]+Configs[j]
  [8] <> 2 then return 0 fi;
    1
  end proc:
> T:= Matrix(256,256,Compatible):
```

Thus $a(n) = u\, T^n\, v$ where u and v are row and column vectors respectively with $u_i = 1$ for i corresponding to configurations with left column $(0, 0, 0, 0)$, 0 otherwise, and $v_i = 1$ for i corresponding to configurations with right column $(0, 0, 0, 0)$, 0 otherwise. The following Maple code produces these vectors.

```
> u:= Vector[row](256, i -> `if`(Configs[i][1..4] = [0,0,0,0],1,0))
  :
  v:= Vector(256, j -> `if`(Configs[j][5..8] = [0,0,0,0],1,0)):
```

To check, here are the first few entries of our sequence.

```
> TV[0]:= v:
  for n from 1 to 15 do TV[n]:= T . TV[n-1] od:
> A:= [seq(u . TV[n],n=1..15)];
```

$$A := [1, 1, 5, 10, 21, 50, 130, 332, 840, 2128, 5408, 13772, 35102, 89465, 227961] \quad \textbf{(1)}$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \mapsto t^{32} - 8\,t^{31} + 28\,t^{30} - 61\,t^{29} + 93\,t^{28} - 105\,t^{27} + 98\,t^{26} - 70\,t^{25} + 45\,t^{24} + 14\,t^{23} - 34\,t^{22} + 81\,t^{21} - 37\,t^{20} + 47\,t^{19} - 10\,t^{18} - 2\,t^{17} + 12\,t^{16} - 70\,t^{15} - 33\,t^{14} - 56\,t^{13} \quad \textbf{(2)}$$

$$-95\,t^{12}-63\,t^{11}-75\,t^{10}-86\,t^{9}-45\,t^{8}-9\,t^{7}$$

This turns out to have degree 32, but with the 7 lowest coefficients 0. Thus we will have $0 = u\,P(T)\,T^n\,v = \sum_{i=7}^{32} p_i\,a(i+n) = \sum_{i=0}^{25} p_{i+7}\,a(i+n-7)$ where p_i is the coefficient of t^i in $P(t)$. That corresponds to a homogeneous linear recurrence of order 25, which would hold true for any u and v, after a delay of 7. It seems that with our particular u and v we have a recurrence of order only 13 with a delay of 3, corresponding to a factor of P.

```
> Q:= unapply((t^13-4*t^12+6*t^11-7*t^10+4*t^9-3*t^8-t^7+5*t^6-4*
  t^5+2*t^4+t^3+5*t+3)*t^3,t);
```

$$Q := t \mapsto \left(t^{13}-4\,t^{12}+6\,t^{11}-7\,t^{10}+4\,t^{9}-3\,t^{8}-t^{7}+5\,t^{6}-4\,t^{5}+2\,t^{4}+t^{3}+5\,t+3\right)t^{3} \qquad \textbf{(3)}$$

The complementary factor $R(t) = \dfrac{P(t)}{Q(t)}$ has degree 16.

```
> .R:= unapply(normal(P(t)/Q(t)),t);
```

$$R := t \mapsto \left(t^{12}-4\,t^{11}+6\,t^{10}-6\,t^{9}+t^{8}-4\,t^{7}-t^{6}-10\,t^{5}-9\,t^{4}-4\,t^{3}-12\,t^{2}-10\,t-3\right)t^{4} \qquad \textbf{(4)}$$

Now we want to show that $b(n) = u\,Q(T)\,T^n\,v = 0$ for all $n \geq 1$. This will certainly satisfy the order-16 recurrence

$$\sum_{i=4}^{16} r_i\,b(i+n) = \sum_{i=4}^{16} r_i u\,Q(T)\,T^{n+i} v = u\,Q(T)\,R(T)\,T^n\,v = u\,P(T)\,T^n\,v = 0$$

where r_i are the coefficients of $R(t)$. To show all $b(n) = 0$ it suffices to show $b(1) = \ldots = b(15) = 0$.

First we compute $w = u\,Q(T)$, then multiply it with the already-computed $T^n v$.

```
> UT[0]:= u:
  for n from 1 to 16 do UT[n]:= UT[n-1].T od:
  w:= add(coeff(Q(t),t,j)*UT[j],j=3..16):
> seq(w . TV[n],n=1..15);
```

$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \qquad \textbf{(5)}$$